MASSACHVSETTS INSTITVTE OF TECHNOLOGY
Department of Electrical Engineering and Computer Science
6.037—Structure and Interpretation of Computer Programs
IAP 2019

Streams

Streams

****** SOLUTIONS ******

Simple Streams:

Zeros: (0 0 0 0 0 0 ....

```
(define zeros (cons-stream 0 zeros)
```

Ones: (1 1 1 1 1 1 ....

```
(define ones (cons-stream 1 ones)
```

Natural numbers (called ints): (1 2 3 4 5 6 ....

```
(define ints (cons-stream 1 (add-streams ones ints)))
```

What happens if you use `cons` instead of `cons-stream`?

Stream operators

We'd like to be able to operate on streams to modify them and combine them with other streams. For example, to do element-wise addition or multiplication:

```
(define (add-streams s1 s2) (map2-stream + s1 s2))
(define (mul-streams s1 s2) (map2-stream * s1 s2))
(define (div-streams s1 s2) (map2-stream / s1 s2))
```

Write map2-stream:

```
(define (map2-stream op s1 s2)
  (cons-stream (op (stream-car s1) (stream-car s2))
               (map2-stream op (stream-cdr s1) (stream-cdr s2))))
```

Another possible operation is multiplying every element of the stream by a constant factor:

```
(define (scale-stream x s)
  (cons-stream (* x (stream-car s))
               (scale-stream x (stream-cdr s))))
```

Implement the stream of factorials, which goes `(1 1 2 6 24 120 ...`:

```
    ints:   1   2   3   4    5     6 ...
 x  facts:  1   1   2   6   24   120 ...
 ------------------------------------------
facts: 1    1   2   6  24  120   720 ...
```

```
(define facts (cons-stream 1 (mul-streams ints facts)))
```

Power Series

We can approximate functions by summing terms of an appropriate power series. A power series has the form:

$$\sum a_n x^n = a_0 + a_1 x + a_2 x^2 + a_3 x^3 + \cdots$$

By selecting appropriate a_n, the series converges to the value of a function. One particularly useful function for which this is the case is e^x which has the following power series:

$$e^x = 0! + \frac{x}{1!} + \frac{x^2}{2!} + \frac{x^3}{3!} + \cdots$$

Since power series involve an infinite summation, of which we might only care about the first couple terms, they are an excellent problem to tackle with streams. The stream will encode the coefficients a_n. For example, to represent the function $f(x) = 3$, we'd use a stream whose first element was 3, and the rest are zeros. The following two procedures come in handy:

```
;; x^0  x^1  x^2  x^3 ...
(define (powers x)
  (cons-stream 1 (scale-stream x (powers x))))

;; sum up the first n terms of the series with coefficients in s, for the given value of x
(define (sum-series s x n)
  (define (sum-helper s sum n)
    (if (= n 0)
        sum
        (sum-helper (stream-cdr s) (+ sum (stream-car s)) (- n 1))))
  (sum-helper (mul-streams s (powers x)) 0 n))
```

Write an expression that computes a stream to represent the power series that converges to $f(x) = 2x + 5$:

```
(define two-x-plus-five (cons-stream 5 (cons-stream 2 zeros)))
```

Write an expression that computes the stream for e^x:

```
(define e-to-the-x (div-streams ones facts))
```

To compute e^5 using 20 terms, we'd call `(sum-series e-to-the-x 5 20)`.

Since the stream represents a function, we can write operations which work on functions and try to implement them in terms of the coefficients of the series. One such operation is integration. The integral of an infinite polynomial is also an infinite polynomial, but the coefficients will be different. In particular, we'll want our integration function to return a stream whose constant term (first element) is missing, as it can't actually compute it from the series. We'll always remember to add a constant term on before using it; the result of integrate-series starts with the co-efficient of x^1, not x^0.

```
;; integral of ax^n is (a/(n+1))x^)n+1
;; We're not going to have an x^0 term (omitting constant)
;;
;;      2       +   3x       +   4x^2
;;->    (2/1)x  +   (3/2)x^2 +   (4/3)x^3

(define (integrate-series s)
  (div-streams s ints))
```

Write a new definition of e^x using `integrate-series` (Hint: what is the integral of e^x?)

```
;; Tack on constant term; we know e^0 is 1
(define e-to-the-x (cons-stream 1 (integrate-series e-to-the-x)))
```

Given that we can build e^x this way, implement sin and cos in a similar fashion:

```
; integral of cosine is sine + C. integral of sine is -cosine + C. sin(0) = 0. cos(0) = 1.
(define sine (cons-stream 0 (integrate-series cosine)))
(define cosine
  (cons-stream 1
    (scale-stream -1 (integrate-series sine))))
```

Bonus Round Problem: Another operation is function multiplication. This involves multiplying two infinite polynomials, which is not the same as `mul-streams`, as that only does elementwise multiplication.

```
; Like doing long multiplication (shown left-to-right instead of right-to-left here.)
;
;      s2:    a   b   c   d   e ...
;  x   s1:    f   g   h   i   j ...
;  ----------------------------------------
;          a*f   f*[b c d e ...]
;              + [a b c ..] * [g h i...]

(define (mul-series s1 s2)
```

```
  (cons-stream (* (stream-car s1) (stream-car s2))
     (add-streams (scale-stream (stream-car s1)
                                (stream-cdr s2))
                  (mul-series (stream-cdr s1) s2))))
```

Then this should look interestingly simple:

```
(add-streams (mul-series sine sine)
             (mul-series cosine cosine))

;
; Yup, sin^2(x) + cos^2(x) = 1 !
;
```